


```

start:                                     ; 程序开始
      mov    a, 0                          ; (1)  ;
      mov    pac, a                        ; 设置PA口为输出口
      mov    pa, a                          ; 清除PA口
; -----
loop:                                     ; LED灯循环
      mov    a, 0                          ;
      mov    tblp, a                       ; 表格指针初始化
      tabrdl rglight                       ; (2)  ; 读取最后一页的表格内容, 放入LED控制寄存器
      mov    a, rglight                    ; (3)  ;
      mov    pa, a                          ; 将LED寄存器的内容输出到PA口
      call   delayl                         ; (4)  ; 调用延时子程序delayl
      inc    tblp                          ; (5)  ; 递增表格指针
      mov    a, 6                          ;
      mov    flash, a                      ; 设置LED闪烁控制寄存器的值为6次
flashlamp:                               ; LED闪烁循环
      tabrdl rglight                       ; 读取最后一页的表格内容, 放入LED控制寄存器
      mov    a, rglight                    ;
      mov    pa, a                          ; 将LED寄存器的内容输出到PA口
      call   delays                         ; (6)  ; 调用延时子程序delays
      inc    tblp                          ; 递增表格指针
      sdz    flash                         ; 检查LED闪烁是否结束
      jmp    flashlamp                     ; 闪烁没有结束, 则跳至LED闪烁循环
      tabrdl rglight                       ; 读取最后一页的表格内容, 放入LED控制寄存器
      mov    a, rglight                    ;
      mov    pa, a                          ; 将LED寄存器的内容输出到PA口
      call   delaym                         ; (7)  ; 调用延时子程序delaym
      inc    tblp                          ; 递增表格指针
; -----
      tabrdl rglight                       ; 读取最后一页的表格内容, 放入LED控制寄存器
      mov    a, rglight                    ;
      mov    pa, a                          ; 将LED寄存器的内容输出到PA口
      call   delayl                         ; 调用延时子程序delayl
      inc    tblp                          ; 递增表格指针
      mov    a, 6                          ; 设置LED闪烁次数
      mov    flash, a                      ; 将闪烁次数存入LED闪烁控制寄存器
flash2:                                  ; LED闪烁循环2
      tabrdl rglight                       ; 读取最后一页的表格内容, 放入LED控制寄存器
      mov    a, rglight                    ;
      mov    pa, a                          ; 将LED寄存器的内容输出到PA口
      call   delays                         ; 调用延时子程序delays
      inc    tblp                          ; 递增表格指针
      sdz    flash                         ; 检查LED闪烁是否结束

```

```

        jmp     flash2           ; 闪烁没有结束，则跳至LED闪烁循环2
tabrdl  rglight                ; 读取最后一页的表格内容，放入LED控制寄存器
        mov     a, rglight      ;
        mov     pa, a           ; 将LED寄存器的内容输出到PA口
        call    delaym         ; 调用延时子程序delaym
        jmp     loop           ;
; -----
delayl proc                       ; 延时子程序delayl
        mov     a, 0fh
        mov     count1, a
        mov     count2, a
        mov     count3, a
d1:     sdz     count3
        jmp     d1
        sdz     count2
        jmp     d1
        sdz     count1
        jmp     d1
        ret
delayl endp
; -----
delaym proc                       ; 延时子程序delaym
        mov     a, 07h
        mov     count1, a
        mov     a, 0ffh
        mov     count2, a
        mov     count3, a
d2:     sdz     count3
        jmp     d2
        sdz     count2
        jmp     d2
        sdz     count1
        jmp     d2
        ret
delaym endp
; -----
delays proc                       ; 延时子程序delays
        mov     a, 0ffh
        mov     count1, a
        mov     count2, a
d3:     sdz     count2
        jmp     d3
        sdz     count1

```

```

        jmp     d3
        ret
delays endp
; -----
        org     300h           ; LED表格
        dc     0ebh, 0fbh, 0ebh, 0fbh, 0ebh, 0fbh, 0ebh, 0dbh
        dc     0beh, 0bfh, 0beh, 0bfh, 0beh, 0bfh, 0beh, 0bdh
end

```

程序说明

程序开始(1)先设置输入/输出控制寄存器，用来确定端口是作为输入还是作为输出使用。在本范例中，设置 PAC 为 0，即 PA 口作为输出口使用。因为红绿灯的点亮条件是固定的，所以我们可以利用查表的方法来确定 PA 口的值，就象(2)所做的那样，而这个表格是由我们事先建立的。TABRDL 指令用来读取最后一页表格内容，这里 ROM 的最大地址是 03FFH，因此最后一页是从 0300H 开始的。在(5)中，表格指针被递增。在(2)中，LED 显示寄存器从表格获得数据，并将这个数据送至输出口进行 LED 显示。由于不同的 LED 点亮的时间有所不同，因此需要不同的延时，程序中总共有三个不同的延时子程序，如程序中所示(4) delay1、(7) delaym、(6) delays。